

# THE INVERSE FISHER TRANSFORM

## By John Ehlers

The purpose of technical indicators is to help with your timing decisions to buy or sell. Hopefully, the signals are clear and unequivocal. However, more often than not your decision to pull the trigger is accompanied by crossing your fingers. Even if you have placed only a few trades you know the drill.

In this article I will show you a way to make your oscillator-type indicators make clear black-or-white indication of the time to buy or sell. I will do this by using the Inverse Fisher Transform to alter the Probability Distribution Function (PDF) of your indicators. In the past<sup>12</sup> I have noted that the PDF of price and indicators do not have a Gaussian, or Normal, probability distribution. A Gaussian PDF is the familiar bell-shaped curve where the long “tails” mean that wide deviations from the mean occur with relatively low probability. The Fisher Transform can be applied to almost any normalized data set to make the resulting PDF nearly Gaussian, with the result that the turning points are sharply peaked and easy to identify. The Fisher Transform is defined by the equation

$$1) \quad y = 0.5 * \ln\left(\frac{1+x}{1-x}\right)$$

Whereas the Fisher Transform is expansive, the Inverse Fisher Transform is compressive. The Inverse Fisher Transform is found by solving equation 1 for x in terms of y. The Inverse Fisher Transform is:

$$2) \quad x = \frac{e^{2y} - 1}{e^{2y} + 1}$$

The transfer response of the Inverse Fisher Transform is shown in Figure 1. If the input falls between -0.5 and +0.5, the output is nearly the same as the input. For larger absolute values (say, larger than 2), the output is compressed to be no larger than unity. The result of using the Inverse Fisher Transform is that the output has a very high probability of being either +1 or -1. This bipolar probability distribution makes the Inverse Fisher Transform ideal for generating an indicator that provides clear buy and sell signals.

---

<sup>1</sup> John Ehlers, “Using the Fisher Transform”, Stocks & Commodities, November 2002, page 40

<sup>2</sup> John Ehlers, “Cybernetic Analysis for Stocks and Futures”, John Wiley, Chapter 1




Figure 1. Transfer Response of the Inverse Fisher Transform Compresses the Output to be Between  $-1$  and  $+1$

One of the more popular technical indicators is a Stochastic RSI. This indicator starts by taking an RSI of price. Then, a Stochastic of that RSI is taken to limit the output to be between 0 and 100. Translating and scaling, this is mathematically the same as varying between  $-1$  and  $+1$ .

Now that you know about the Inverse Fisher Transform, there is no reason to bludgeon the RSI with a blunt instrument like a Stochastic. Instead of picking an observation length that is guaranteed to drive the Stochastic to saturation, you can finesse the indicator PDF using the Inverse Fisher Transform. The EasyLanguage code to do this is given in Figure 2. The 5 bar RSI varies from a minimum of 0 and a maximum of 100. The 5 bar length of the RSI was selected to provide good operation when applied to many price series. The RSI period is certainly available for optimization. By subtracting 50, the RSI indicator is translated to range from  $-50$  to  $+50$ . Then, multiplying by 0.1 reduces the range to be between  $-5$  and  $+5$  for Value1. This is just the kind of maximum swing suited for the Inverse Fisher Transform. I used a 9 bar weighted moving average to compute Value2 to smooth Value1 and ultimately remove some spurious trading signals. There is no magic in this average. It could have fewer bars to have less lag or it could be an Exponential Moving Average. Its function is just to be a smoother. The transform is calculated as the variable IFish and then plotted. The code also plots output reference lines at  $-0.5$  and  $+0.5$ .

```

Vars: IFish(0);

Value1 = .1*(RSI(Close, 5) - 50);
Value2 = WAverage(Value1, 9);
IFish = (ExpValue(2*Value2) - 1) / (ExpValue(2*Value2) + 1);

Plot1(IFish, "IFish");
Plot2(0.5, "Sell Ref");
Plot3(-0.5, "Buy Ref");

```

Figure 2. EasyLanguage Code to Take the Inverse Fisher Transform of an RSI

The transformed RSI is applied to the Exchange Traded Fund (ETF) QQQ in Figure 3. I demonstrate the Inverse Fisher Transform using ETFs because they can be bought long or sold short with equal facility – just like Futures. The trading rules are simple. Buy when the indicator crosses over  $-0.5$  or crosses over  $+0.5$  if it has not previously crossed over  $-0.5$  and to sell short when the indicators crosses under  $+0.5$  or crosses under  $-0.5$  if it has not previously crossed under  $+0.5$ . You can see that the trading signals are not only clear and unequivocal, but they are also profitable.



Figure 3. Inverse Fisher RSI Clear and Profitable Signals

The use of the Inverse Fisher Transform is not limited to just altering the RSI PDF. It can be applied to almost any oscillator-type indicator. For example, my simplified model of the market consists of a trend component and a cycle

component. The cycle component can be isolated by filtering. I call it the Cyber Cycle. Like the RSI, the Cyber Cycle is an oscillator-type indicator. Unlike the RSI, the Cyber Cycle has cyclic swings with variable amplitude. Ensuring the cyclic swings of the Cyber Cycle has sufficient amplitude to allow the Inverse Fisher Transform to invoke its compression, an excellent indicator can result.

For example, Figure 4 is the EasyLanguage code for the Cyber Cycle followed by the Inverse Fisher Transform.

```
Inputs: Price((H+L)/2),
        alpha(.07);

Vars: Smooth(0),
      Cycle(0),
      ICycle(0);

Smooth = (Price + 2*Price[1] + 2*Price[2] + Price[3])/6;
Cycle = (1 - .5*alpha)*(1 - .5*alpha)*(Smooth - 2*Smooth[1] + Smooth[2]) + 2*(1 -
alpha)*Cycle[1] - (1 - alpha)*(1 - alpha)*Cycle[2];
If currentbar < 7 then Cycle = (Price - 2*Price[1] + Price[2]) / 4;

ICycle = (ExpValue(2*Cycle) - 1) / (ExpValue(2*Cycle) + 1);

Plot1(ICycle, "Cycle");
Plot2(0.5, "Sell Ref");
Plot3(-0.5, "Buy Ref");
```

Figure 4. EasyLanguage Code for the Cyber Cycle with the Inverse Fisher Transform

The pure Cyber Cycle indicator for the SPY ETF is shown in the first subgraph of Figure 5. The variable amplitude cyclic swings are obvious. One can trade the Cyber Cycle using the crossing of the indicator and the indicator delayed by one bar. The transformed result is shown in the second subgraph of Figure 5. As with the Transformed RSI, the buy and sell signals are clear and unambiguous. The Inverse Fisher Transform can be applied with equal success to virtually all oscillator-type indicators.




Figure 5. Good Trading Signals Arise from Applying the Inverse Fisher Transform to the Cyber Cycle Indicator

The Inverse Fisher Transform has even wider potential applications. Since the transformed waveform is limited to the range between  $-1$  and  $+1$ , total energy in the wave is limited. I am particularly intrigued that convergence is guaranteed in some linear predictive algorithms when the energy in the wave is limited. Thus research may reveal still more exciting new results for traders.

More importantly, for the present, I have shown how using the Inverse Fisher Transform can let you have greater confidence (and perhaps uncross your fingers) when you place your trades.